// File: lib/openzeppelin-contracts/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/cryptography/ECDSA.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/ECDSA.sol)

pragma solidity ^0.8.0;

import "../Strings.sol";

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV // Deprecated in v4.8
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address, RecoverError) {
        bytes32 s = vs & bytes32(0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff);
        uint8 v = uint8((uint256(vs) >> 255) + 27);
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from `s`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes memory s) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n", Strings.toString(s.length), s));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}

// File: src/interfaces/payments/IDopaminePrimaryCheckout.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IDopaminePrimaryCheckoutEventsAndErrors} from "./IDopaminePrimaryCheckoutEventsAndErrors.sol";
import {IOwnable} from "../utils/IOwnable.sol";
import {Bundle, Order} from "../../payments/DopaminePrimaryCheckoutStructs.sol";
import {IEIP712Signable} from "../utils/IEIP712Signable.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Dopamine Primary Checkout Interface
interface IDopaminePrimaryCheckout is IOwnable, IEIP712Signable, IDopaminePrimaryCheckoutEventsAndErrors {

    /// @notice Derives an order hash for a set of bundle purchases.
    function getOrderHash(
        string memory id,
        address purchaser,
        Bundle[] calldata bundles
    ) external returns (bytes32);

    /// @notice Derives the bundle hash associated with a specific bundle.
    function getBundleHash(
        uint64 brand,
        uint64 collection,
        uint64 colorway,
        uint64 size,
        uint256 price
    ) external returns (bytes32);

    /// @notice Processes a Dopamine bundle order.
    function checkout(Order calldata order) external payable;

    /// @notice Withdraw `amount` in wei from the contract address.
    function withdraw(uint256 amount, address to) external;

    /// @notice Sets an approved order signer for the contract.
    function setSigner(address signer, bool setting) external;

}

// File: src/interfaces/payments/IDopaminePrimaryCheckoutEventsAndErrors.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {Bundle} from "../../payments/DopaminePrimaryCheckoutStructs.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Dopamine Primary Checkout Events & Errors Interface
interface IDopaminePrimaryCheckoutEventsAndErrors {

    /// @notice Emits when new checkout signer permissions are set.
    /// @param signer The address whose signing permissions are being set.
    /// @param setting Whether or not the address may verify checkouts.
    event DopamineCheckoutSignerSet(address signer, bool setting);

    /// @notice Emits when an order is successfully fulfilled.
    /// @param orderHash The tracking identifier of the Dopamine order.
    /// @param purchaser The address who completed the checkout.
    /// @param bundles All bundled items included with the Dopamine order.
    event OrderFulfilled(
        bytes32 orderHash,
        address indexed purchaser,
        Bundle[] bundles
    );

    /// @notice The provided order has already been processed.
    error OrderAlreadyProcessed();

    /// @notice Error when transferring ETH to the sender.
    error EthTransferFailed();

    /// @notice The number of bundles ordered exceeds the maximum allowed.
    error OrderCapacityExceeded();

    /// @notice Insufficient payment provided for the purchase.
    error PaymentInsufficient();

    /// @notice Amount specified for withdrawal is invalid.
    error WithdrawalInvalid();

}

// File: src/interfaces/utils/IEIP712Signable.sol



pragma solidity ^0.8.18;

import {IEIP712SignableErrors} from "./IEIP712SignableErrors.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title EIP-712 Signable Interface
interface IEIP712Signable is IEIP712SignableErrors {

    /// @notice Returns all EIP-712 metadata pertaining to the contract.
    function EIP712Data() external view returns (
        string memory name,
        string memory version,
        address verifyingContract,
        bytes32 domainSeparator
    );

}

// File: src/interfaces/utils/IEIP712SignableErrors.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title EIP712 Signable Errors Interface
interface IEIP712SignableErrors {

    /// @notice The provided signature has expired.
    error SignatureExpired();

    /// @notice The provided signature is not valid.
    error SignatureInvalid();

}

// File: src/interfaces/utils/IERC173.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IERC165} from "@openzeppelin/contracts/utils/introspection/IERC165.sol";

import {IERC173Events} from "./IERC173Events.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title IERC173 Interface
interface IERC173 is IERC165, IERC173Events {

    /// @notice Get the owner address of the contract.
    /// @return The address of the owner.
    function owner() view external returns(address);

    /// @notice Set the new owner address of the contract.
    function transferOwnership(address newOwner) external;

}

// File: src/interfaces/utils/IERC173Events.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title IERC173 Events & Errors Interface
interface IERC173Events {

    /// @notice Emits when ownership is transferred to another address.
    /// @param previousOwner The address of the prior owner.
    /// @param newOwner The address of the new owner.
    event OwnershipTransferred(
        address indexed previousOwner,
        address indexed newOwner
    );

}

// File: src/interfaces/utils/IOwnable.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IERC173} from "./IERC173.sol";
import {IOwnableEventsAndErrors} from "./IOwnableEventsAndErrors.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Ownable Interface
interface IOwnable is IERC173, IOwnableEventsAndErrors {

    /// @notice Gets the pending owner of the contract.
    /// @return The pending owner address for the contract.
    function pendingOwner() external returns(address);

    /// @notice Sets the pending owner address for the contract.
    /// @param pendingOwner The address of the new pending owner.
    function setPendingOwner(address pendingOwner) external;

    /// @notice Permanently renounces contract ownership.
    function renounceOwnership() external;

}

// File: src/interfaces/utils/IOwnableEventsAndErrors.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Ownable Events & Errors Interface
interface IOwnableEventsAndErrors {

    /// @notice Emits when a new pending owner is set.
    /// @param pendingOwner The address of the new pending owner.
    event PendingOwnerSet(
        address indexed pendingOwner
    );

    /// @notice Caller is not the owner of the contract.
    error OwnerOnly();

    /// @notice The pending owner is invalid.
    error PendingOwnerInvalid();

    /// @notice Caller is not the pending owner of the contract.
    error PendingOwnerOnly();

    /// @notice The pending owner is already set to the specified address.
    error PendingOwnerAlreadySet();

}

// File: src/interfaces/utils/IReentrancyGuardErrors.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Reentrancy Guard Errors Interface
interface IReentrancyGuardErrors {

    /// @notice Function reentrancy was detected.
    error FunctionReentrant();

}

// File: src/payments/DopaminePrimaryCheckout.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IERC165} from "@openzeppelin/contracts/utils/introspection/IERC165.sol";

import {IERC173} from "../interfaces/utils/IERC173.sol";
import {IDopaminePrimaryCheckout} from "../interfaces/payments/IDopaminePrimaryCheckout.sol";

import {Ownable} from "../utils/Ownable.sol";
import {ReentrancyGuard} from "../utils/ReentrancyGuard.sol";
import {EIP712Signable} from "../utils/EIP712Signable.sol";
import {Order, Bundle} from "./DopaminePrimaryCheckoutStructs.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title  Dopamine Primary Checkout
/// @author @leeren
/// @notice Crypto checkout attribution for Dopamine Streetwear+ bundle orders.
contract DopaminePrimaryCheckout is ReentrancyGuard, Ownable, EIP712Signable, IDopaminePrimaryCheckout {

    /// @notice Do not permit more than 10 bundles to be purchased at a time.
    uint256 MAX_BUNDLES = 10;

    /// @notice EIP-712 claim hash used for processing and validating orders.
    bytes32 public constant ORDER_TYPEHASH = keccak256("Order(string id,address purchaser,Bundle[] bundles)Bundle(uint64 brand,uint64 collection,uint64 colorway,uint64 size,uint256 price)");
    bytes32 public constant BUNDLE_TYPEHASH = keccak256("Bundle(uint64 brand,uint64 collection,uint64 colorway,uint64 size,uint256 price)");

    /// @notice Tracks whether a bundle was purchased or not.
    mapping(string => bool) public orderProcessed;

    /// @notice Instantiates the Dopamine Primary Checkout contract.
    /// @param signer Address whose sigs are used for purchase verification.
	constructor(address signer) {
        setSigner(signer, true);
    }

    /// @notice Gets the EIP-712 order hash or tracking crypto-based purchases.
    /// @param id The Dopamine tracking identifier associated with the order.
    /// @param purchaser The address granted permission to make the purchase.
    /// @param bundles A list of bundled products included in the order.
    /// @return The EIP-712 hash identifying the marketplace order.
    function getOrderHash(
        string memory id,
        address purchaser,
        Bundle[] calldata bundles
    ) public pure returns (bytes32) {
        bytes32[] memory bundleHashes = new bytes32[](bundles.length);
        Bundle memory bundle;
        for (uint256 i = 0; i < bundles.length; ++i) {
            bundle = bundles[i];
            bundleHashes[i] = getBundleHash(
                bundle.brand, bundle.collection, bundle.colorway, bundle.size, bundle.price
            );
        }
        return keccak256(
            abi.encode(
                ORDER_TYPEHASH,
                keccak256(bytes(id)),
                purchaser,
                keccak256(abi.encodePacked(bundleHashes))
            )
        );
    }

    /// @notice Gets the EIP-712 bundle hash for a specific merchandise item.
    /// @param brand The identifier for the bundle brand.
    /// @param collection The product identifier for a bundle.
    /// @param colorway The chosen bundle colorway identifier.
    /// @param size The size identifier for a bundle.
    /// @param price The price (in USD) of the individual bundle.
    /// @return The EIP-712 hash identifying the bundle.
    function getBundleHash(
        uint64 brand,
        uint64 collection,
        uint64 colorway,
        uint64 size,
        uint256 price
    ) public pure returns (bytes32) {
        return keccak256(
            abi.encode(
                BUNDLE_TYPEHASH,
                brand,
                collection,
                colorway,
                size,
                price
            )
        );
    }

    /// @notice Performs registered checkout of a crypto-based Dopamine order.
    /// @param order The bundle order to be processed.
    function checkout(Order calldata order) external payable nonReentrant {
        if (orderProcessed[order.id]) {
            revert OrderAlreadyProcessed();
        }
        if (order.bundles.length > MAX_BUNDLES) {
            revert OrderCapacityExceeded();
        }

        (bytes32 orderHash) = getOrderHash(
            order.id,
            order.purchaser,
            order.bundles
        );

        _verifySignature(
            order.signature,
            _deriveEIP712Digest(orderHash)
        );

        orderProcessed[order.id] = true;
        emit OrderFulfilled(orderHash, order.purchaser, order.bundles);
    }

    /// @notice Withdrawals all collected payments to the owner address.
    /// @param amount The amount in wei to withdraw.
    /// @param to The address to withdraw the wei to.
    function withdraw(uint256 amount, address to) public onlyOwner {
        if (amount > address(this).balance) {
            revert WithdrawalInvalid();
        }
        _transferETH(payable(to), amount);
    }

    /// @notice Sets up a new signer for authorizing checkouts.
    /// @param signer The address verifying crypto-based checkouts.
    /// @param setting Whether the signer has permission to verify checkouts.
    function setSigner(address signer, bool setting) public onlyOwner {
        signers[signer] = setting;
        emit DopamineCheckoutSignerSet(signer, setting);
    }

    /// @notice Checks if the given contract supports an interface.
    /// @param interfaceId The 4-byte ERC-165 interface identifier.
    function supportsInterface(bytes4 interfaceId) public override(IERC165, Ownable) pure returns (bool) {
        return interfaceId == type(IERC165).interfaceId ||
               interfaceId == type(IERC173).interfaceId;
    }

    /// @notice Transfers `amount` wei to address `to`.
    /// @param to The address receiving the transfer.
    /// @param amount The amount in wei being transferred.
    function _transferETH(address payable to, uint256 amount) public {
        (bool success, ) = to.call{value: amount}("");
        if(!success){
            revert EthTransferFailed();
        }
    }

    /// @notice EIP-712 name identifier used for signature verification.
    function _NAME() internal override pure returns (string memory) {
        return "Dopamine Primary Checkout";
    }

    /// @notice EIP-712 version identifier used for signature verification.
    function _VERSION() internal override pure returns (string memory) {
        return "1.0";
    }

}

// File: src/payments/DopaminePrimaryCheckoutStructs.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @notice Encapsulation of a Dopamine Marketplace order, which is composed of:
///         - id, a string UUID for uniquely identifying a Dopamine order
///         - purchaser, the address of the entity processing a crypto checkout
///         - signature, a Dopamine signature used to attest order validity
///         - bundles, a list of all Dopamine bundles included in the order
struct Order {
    string id;
    address purchaser;
    bytes signature;
    Bundle[] bundles;
}

/// @notice Encapsulation of a Dopamine Marketplace bundle, which comprises of:
///         - brand, a uint64 identifier for brand attribution (e.g. Nouns)
///         - collection, a uint64 identifier for the collection (e.g. Hoodie)
///         - colorway, a uint64 identifier for the colorway (e.g. Arctic Lime)
///         - size, a uint64 identifier for the bundle's physical (e.g. XL)
///         - price, a uint256 identifier for the bundle's price in USD
struct Bundle {
    uint64 brand;
    uint64 collection;
    uint64 colorway;
    uint64 size;
    uint256 price;
}

// File: src/utils/EIP712Signable.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {ECDSA} from "@openzeppelin/contracts/utils/cryptography/ECDSA.sol";
import {IEIP712Signable} from "../interfaces/utils/IEIP712Signable.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title EIP-712 Signable Contract
abstract contract EIP712Signable is IEIP712Signable {

    uint256 internal immutable _CHAIN_ID;
    bytes32 internal immutable _DOMAIN_SEPARATOR;

    /// @notice Gets whether an address is an authorized signer.
    mapping(address => bool) public signers;

    constructor() {
        _CHAIN_ID = block.chainid;
        _DOMAIN_SEPARATOR = _buildDomainSeparator();
    }

    function _NAME() virtual internal pure returns (string memory);

    function _VERSION() virtual internal pure returns (string memory);

    /// @notice Returns all EIP-712 metadata pertaining to the contract.
    function EIP712Data() external view 
        returns (
            string memory name,
            string memory version,
            address verifyingContract,
            bytes32 domainSeparator
        )
    {
        name = _NAME();
        version = _VERSION();
        verifyingContract = address(this);
        domainSeparator = _buildDomainSeparator();
    }

    /// @notice Derives the contract EIP-712 digest of a given hash.
    function _deriveEIP712Digest(bytes32 hash) internal view returns (bytes32) {
        return ECDSA.toTypedDataHash(_domainSeparator(), hash);
    }

    /// @notice Performs signatory verification of a signature and digest pair.
    function _verifySignature(bytes memory signature, bytes32 digest, address signer) internal pure {
        address signatory = ECDSA.recover(digest, signature);
        if (signatory == address(0) || signatory != signer) {
            revert SignatureInvalid();
        }
    }

    /// @notice Performs signatory verification of a signature and digest pair.
    function _verifySignature(bytes memory signature, bytes32 digest) internal view {
        address signatory = ECDSA.recover(digest, signature);
        if (signatory == address(0) || !signers[signatory]) {
            revert SignatureInvalid();
        }
    }

    /// @dev Generates an EIP-712 domain separator.
    /// @return A 256-bit domain separator tied to this contract.
    function _buildDomainSeparator() internal view returns (bytes32) {
        return keccak256(
            abi.encode(
                keccak256("EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"),
                keccak256(bytes(_NAME())),
                keccak256(bytes(_VERSION())),
                block.chainid,
                address(this)
            )
        );
    }

    /// @dev Returns the domain separator tied to the contract.
    /// @return 256-bit domain separator tied to this contract.
    function _domainSeparator() internal view returns (bytes32) {
        if (block.chainid == _CHAIN_ID) {
            return _DOMAIN_SEPARATOR;
        } else {
            return _buildDomainSeparator();
        }
    }

}

// File: src/utils/Ownable.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IOwnable} from "../interfaces/utils/IOwnable.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

/// @title Ownable Contract
abstract contract Ownable is IOwnable {

    /// @notice Used for permanently revoking ownership.
    address public constant REVOKE_OWNER_ADDRESS = 0x000000000000000000000000000000000000dEaD;

    address public owner;
    address public pendingOwner;

    modifier onlyOwner() {
        if (msg.sender != owner) {
            revert OwnerOnly();
        }
        _;
    }

    modifier onlyPendingOwner() {
        if (msg.sender != pendingOwner) {
            revert PendingOwnerOnly();
        }
        _;
    }

    constructor() {
        owner = msg.sender;
    }

    function setPendingOwner(address newPendingOwner) public onlyOwner {
        if (pendingOwner == newPendingOwner) {
            revert PendingOwnerAlreadySet();
        }
        pendingOwner = newPendingOwner;
        emit PendingOwnerSet(newPendingOwner);
    }

    function renounceOwnership() public onlyOwner {
        if (pendingOwner != REVOKE_OWNER_ADDRESS) {
            revert PendingOwnerInvalid();
        }
        _transferOwnership(address(0));
    }

    function transferOwnership(address newOwner) public onlyPendingOwner {
        if (pendingOwner != newOwner) {
            revert PendingOwnerInvalid();
        }
        _transferOwnership(newOwner);
    }

    function supportsInterface(bytes4 id) public view virtual returns (bool);

    /// @notice Transfers ownership to address `newOwner`.
    function _transferOwnership(address newOwner) internal {
        address oldOwner = owner;
        pendingOwner = address(0);
        owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }

}

// File: src/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.18;

import {IReentrancyGuardErrors} from "../interfaces/utils/IReentrancyGuardErrors.sol";

////////////////////////////////////////////////////////////////////////////////
///              ░▒█▀▀▄░█▀▀█░▒█▀▀█░█▀▀▄░▒█▀▄▀█░▄█░░▒█▄░▒█░▒█▀▀▀              ///
///              ░▒█░▒█░█▄▀█░▒█▄▄█▒█▄▄█░▒█▒█▒█░░█▒░▒█▒█▒█░▒█▀▀▀              ///
///              ░▒█▄▄█░█▄▄█░▒█░░░▒█░▒█░▒█░░▒█░▄█▄░▒█░░▀█░▒█▄▄▄              ///
////////////////////////////////////////////////////////////////////////////////

abstract contract ReentrancyGuard is IReentrancyGuardErrors {

    uint256 private constant _LOCKED = 1;
    uint256 private constant _UNLOCKED = 2;

    uint256 internal _locked = _UNLOCKED;

    modifier nonReentrant() {
        if (_locked != _UNLOCKED) {
            revert FunctionReentrant();
        }
        _locked = _LOCKED;
        _;
        _locked = _UNLOCKED;
    }

}